"DroidNim.bas", listed 06-16-2004

   10 CLS:KEY OFF:DEFINT X,Y,Z:DEFSTR Q:Q=MKI$(0):RANDOMIZE TIMER:GOTO 4170 ' droidnim.bas

   20 MD$ = "Ridiculous":RETURN

   30 MD$ = "Absurd":RETURN

   40 MD$ = "Grotesque":RETURN

   50 MD$ = "Nonsensical":RETURN

   60 MD$ = "Farcical":RETURN

   70 MD$ = "Preposterous":RETURN

   80 MD$ = "Silly":RETURN

   90 MD$ = "Senseless":RETURN

  100 MD$ = "Irrational":RETURN

  110 MD$ = "Fantastic":RETURN

  120 MD$ = "Odd":RETURN

  130 MD$ = "Rude":RETURN

  140 MD$ = "Brutish":RETURN

  150 MD$ = "Barbaric":RETURN

  160 MD$ = "Plebeian":RETURN

  170 MD$ = "Uncivil":RETURN

  180 MD$ = "Discourteous":RETURN

  190 MD$ = "Vulgar":RETURN

  200 MD$ = "Coarse":RETURN

  210 MD$ = "Gross":RETURN

  220 MD$ = "Ungraceful":RETURN

  230 MD$ = "Monstrous":RETURN

  240 MD$ = "Horrid":RETURN

  250 MD$ = "Shocking":RETURN

  260 MD$ = "Cheap":RETURN

  270 GOTO 680  ' clear screen to the right of executioners

  280 GOTO 720  ' first move

  290 GOTO 1180 ' select random 1 - 7

  300 GOTO 1420 ' binary select 0 to 111

  310 GOTO 1520 ' you win

  320 GOTO 1700 ' select a response

  330 GOTO 2340 ' print executioners

  340 GOTO 2370 ' animate them

  350 GOTO 2400 ' executioner nods NO

  360 GOTO 2430 ' turn to the left

  370 GOTO 2570 ' look up

  380 GOTO 2580 ' look down

  390 GOTO 2600 ' executioner nods YES

  400 GOTO 2640 ' shooting hand

  410 GOTO 2690 ' pistol

  420 GOTO 2760 ' rh down

  430 GOTO 2790 ' animate androids

  440 GOTO 2850 ' select from 18 animations

  450 GOTO 3180 ' delay 530/10000

  460 GOTO 3190 ' delay 430/10000

  470 GOTO 3200 ' delay 330/10000

  480 GOTO 3230 ' loop 3 times

  490 GOTO 3560 ' clear score

  500 GOTO 3630 ' row 1 executioner agree and shoots

  510 GOTO 3640 ' row 2 executioner agree and shoots

  520 GOTO 3650 ' row 3 executioner agree and shoots

  530 GOTO 3670 ' faces turn to the left and look up or down

  540 GOTO 3810 ' shoot androids

  550 GOTO 4100 ' print @ to locate converter and color androids

  560 GOTO 4810 ' wait for key

  570 GOTO 4840 ' wait for key and animate

  580 ' Draw Executioners

  590 CLS:E=25:GOSUB 550:PRINT CHR$(34);"ANDROID NIM";CHR$(34)

  600 X=194:GOSUB 330:X=581:GOSUB 330:X=963:GOSUB 330

  610 Y9=1:RW$(1)="0":RW$(2)="0":RW$(3)="0":GOSUB 270:GOTO 710

  620 ' Draw Androids

  630 X=254 :FOR N=1 TO 7:X=X-7:GOSUB 330:NEXT

  640 X=634 :FOR N=1 TO 5:X=X-8:GOSUB 330:NEXT

  650 X=1013:FOR N=1 TO 3:X=X-9:GOSUB 330:NEXT

  660 RW$(1)="7":RW$(2)="5":RW$(3)="3":GOTO 3290 ' X9=2 computer plays first

  670 ' Clear Screen to the right of executioners

  680 FOR N=77 TO 1101 STEP 64:E=N:GOSUB 550:PRINT STRING$(50,32):NEXT

  690 E=22:GOSUB 550:PRINT " ":RETURN

  700 ' First move

  710 E=540:GOSUB 550:PRINT "First move by you=1 or me=2 ?":GOTO 740

  720 X=194:GOSUB 340:X=581:GOSUB 340:X=963:GOSUB 340:RETURN ' Winner looks to the right

  730 ' Scan keys

  740 GOSUB 570:EK=ASC(Q):IF ASC(Q)=27 THEN 4910 ' exits game

  750 IF EK=49 THEN 820 ELSE IF EK=50 THEN 780

  760 GOTO 740

  770 ' Computer starts first

  780 X9=2:E=540:GOSUB 550:PRINT "OK, I'll start !             " :'OK

  790 FOR O=1 TO 12:GOSUB 430:NEXT

  800 E=540:GOSUB 550:PRINT      "                             ":GOTO 630

  810 ' Player starts first

  820 X9=1:E=540:GOSUB 550:PRINT "Very well, you may start !   "

  830 X=581:GOSUB 350:X=963:GOSUB 350:X=194:GOSUB 350:GOTO 790

  840 ' My turn

  850 E=22:GOSUB 550:PRINT " "

  860 ON Y9 GOTO 880,870,870

  870 M=INT(RND*2)+1:ON M GOTO 880,900

  880 FOR O=1 TO 12:GOSUB 430:NEXT:GOTO 1130

  890 ' Computer's choice

  900 Z=1:GOSUB 300:Z1=Z0:Z=2:GOSUB 300:Z2=Z0:Z=3:GOSUB 300:Z3=Z0

  910 Z4=Z1+Z2:Z5=Z1+Z3:Z6=Z2+Z3:Z4$=STR$(Z4):Z5$=STR$(Z5):Z6$=STR$(Z6)

  920 GOSUB 430:IF Z4=0 THEN 960

  930 Z0=LEN(Z4$):Y0=1:Z4=0:FOR N=Z0 TO 1 STEP -1

  940 IF MID$(Z4$,N,1)="1" THEN Z4=Z4+Y0

  950 Y0=Y0*2:NEXT

  960 GOSUB 430:IF Z5=0 THEN 1000

  970 Z0=LEN(Z5$):Y0=1:Z5=0:FOR N=Z0 TO 1 STEP -1

  980 IF MID$(Z5$,N,1)="1" THEN Z5=Z5+Y0

  990 Y0=Y0*2:NEXT

 1000 GOSUB 430:IF Z6=0 THEN 1060

 1010 Z0=LEN(Z6$):Y0=1:Z6=0:FOR N=Z0 TO 1 STEP -1

 1020 IF MID$(Z6$,N,1)="1" THEN Z6=Z6+Y0

 1030 Y0=Y0*2:NEXT

 1040 GOSUB 430

 1050 ' Row Indicator

 1060 IF VAL(RW$(3))<=Z4 THEN 1080

 1070 RW=3:GOSUB 520:RN=VAL(RW$(3))-Z4:RN$=STR$(RN):RN$=RIGHT$(RN$,1):X9=1:GOTO 3440

 1080 IF VAL(RW$(2))<=Z5 THEN 1100

 1090 RW=2:GOSUB 510:RN=VAL(RW$(2))-Z5:RN$=STR$(RN):RN$=RIGHT$(RN$,1):X9=1:GOTO 3440

 1100 IF VAL(RW$(1))<=Z6 THEN 1120

 1110 RW=1:GOSUB 500:RN=VAL(RW$(1))-Z6:RN$=STR$(RN):RN$=RIGHT$(RN$,1):X9=1:GOTO 3440

 1120 GOSUB 430:X8=0:IF (Z4=0) AND (Z5=0) THEN 1240 ' end of game

 1130 M=INT(RND*3)+1

 1140 ON M GOTO 1150,1160,1170

 1150 IF VAL(RW$(1))=0 THEN 1160 ELSE RW=1:GOSUB 290:GOSUB 500:X9=1:X8=0:GOTO 3440

 1160 IF VAL(RW$(2))=0 THEN 1170 ELSE RW=2:GOSUB 290:GOSUB 510:X9=1:X8=0:GOTO 3440

 1170 IF VAL(RW$(3))=0 THEN 1150 ELSE RW=3:GOSUB 290:GOSUB 520:X9=1:X8=0:GOTO 3440

 1180 M=INT(RND*7)+1:ON M GOTO 1190,1190,1190,1190,1200,1210,1220

 1190 RN=1:RN$="1":RETURN

 1200 RN=2:RN$="2":RETURN

 1210 RN=3:RN$="3":RETURN

 1220 IF RW=3 THEN 1210 ELSE RN=4:RN$="4":RETURN

 1230 ' There is a win

 1240 GOSUB 270           ' clear right of executioners

 1250 IF X9=1 THEN 1320   ' computer wins

 1260 GOSUB 280

 1270 GOSUB 310:GOSUB 430 ' player wins

 1280 X=194:GOSUB 390     ' yes

 1290 X=963:GOSUB 390

 1300 X=581:GOSUB 390:GOTO 1350

 1310 ' Computer wins

 1320 GOSUB 280

 1330 E=536:GOSUB 550:PRINT MB$:E=472:GOSUB 550:PRINT MA$:E=600:GOSUB 550:PRINT MC$

 1340 GOSUB 470

 1350 E=794:GOSUB 550:PRINT "NEW GAME?  yes <1> or no <2> ?"

 1360 GOSUB 570:IF ASC(Q)=49 THEN 610

 1370 GOTO 4910 ' exit

 1380 GOSUB 270:RW$(1)="0":RW$(2)="0":RW$(3)="0"

 1390 E=540:GOSUB 550:PRINT "Since you have given up ... "

 1400 FOR O=1 TO 8:GOSUB 430:NEXT:GOSUB 270:GOTO 1320

 1410 ' Binary select 000...111

 1420 ON VAL(RW$(Z))+1 GOTO 1430,1440,1450,1460,1470,1480,1490,1500

 1430 Z0=0  :RETURN

 1440 Z0=1  :RETURN

 1450 Z0=10 :RETURN

 1460 Z0=11 :RETURN

 1470 Z0=100:RETURN

 1480 Z0=101:RETURN

 1490 Z0=110:RETURN

 1500 Z0=111:RETURN

 1510 ' You Win

 1520 GOSUB 320:W1$=MD$

 1530 GOSUB 320:IF MD$=W1$ THEN 1530 ELSE W2$=MD$

 1540 GOSUB 320:IF (MD$=W1$) OR (MD$=W2$) THEN 1540 ELSE W3$=MD$

 1550 GOSUB 320:IF (MD$=W1$) OR (MD$=W2$) OR (MD$=W3$) THEN 1550 ELSE W4$=MD$

 1560 GOSUB 320:IF (MD$=W1$) OR (MD$=W2$) OR (MD$=W3$) OR (MD$=W4$) THEN 1560 ELSE W5$=MD$

 1570 GOSUB 320:IF (MD$=W1$) OR (MD$=W2$) OR (MD$=W3$) OR (MD$=W4$) OR (MD$=W5$) THEN 1570 

        ELSE W6$=MD$

 1580 GOSUB 320:IF (MD$=W1$) OR (MD$=W2$) OR (MD$=W3$) OR (MD$=W4$) OR (MD$=W5$) OR (MD$=W6$) 

        THEN 1580 ELSE W7$=MD$

 1590 E=412:GOSUB 550:PRINT "Through some ";W1$+", "

 1600 GOSUB 480:E=470:GOSUB 550:PRINT W2$+", "

 1610 GOSUB 480:E=470:GOSUB 550:PRINT W2$+", "+W3$+", "

 1620 GOSUB 480:E=470:GOSUB 550:PRINT W2$+", "+W3$+", "+W4$+", "

 1630 GOSUB 480:E=534:GOSUB 550:PRINT W5$+", "

 1640 GOSUB 430:E=534:GOSUB 550:PRINT W5$+", "+W6$+", "

 1650 GOSUB 430:E=534:GOSUB 550:PRINT W5$+", "+W6$+", "+W7$+", "

 1660 GOSUB 430:E=610:GOSUB 550:PRINT "Stroke of fate..."

 1670 GOSUB 480:E=678:GOSUB 550:PRINT "YOU WIN !"

 1680 RETURN

 1690 ' Select a response

 1700 M=INT(RND*17)+1

 1710 ON M GOSUB 

        20,30,40,50,60,70,80,90,100,110,120,130,140,150,160,170,180,190,200,210,220,230,240,

        250,260

 1720 RETURN

 1730 ' Select Android's position

 1740 IF VAL(RW$(1))<7 THEN 1750 ELSE X=205:RETURN    ' row 1  10  11 12 13 14 15 16 17

 1750 IF VAL(RW$(1))<6 THEN 1760 ELSE X=212:GOTO 1920

 1760 IF VAL(RW$(1))<5 THEN 1770 ELSE X=219:RETURN

 1770 IF VAL(RW$(1))<4 THEN 1780 ELSE X=226:RETURN

 1780 IF VAL(RW$(1))<3 THEN 1790 ELSE X=233:GOTO 1990

 1790 IF VAL(RW$(1))<2 THEN 1800 ELSE X=240:RETURN

 1800 IF VAL(RW$(1))<1 THEN 1810 ELSE X=247:RETURN

 1810 X=194:RETURN

 1820 IF VAL(RW$(2))<5 THEN 1830 ELSE X=594:RETURN    ' row 2  20  21 22 23 24 25

 1830 IF VAL(RW$(2))<4 THEN 1840 ELSE X=602:GOTO 2050

 1840 IF VAL(RW$(2))<3 THEN 1850 ELSE X=610:GOTO 2140

 1850 IF VAL(RW$(2))<2 THEN 1860 ELSE X=618:RETURN

 1860 IF VAL(RW$(2))<1 THEN 1870 ELSE X=626:RETURN

 1870 X=581:RETURN

 1880 IF VAL(RW$(3))<3 THEN 1890 ELSE X=986 :RETURN   ' row 3  30  31 32 33

 1890 IF VAL(RW$(3))<2 THEN 1900 ELSE X=995 :GOTO 2200

 1900 IF VAL(RW$(3))<1 THEN 1910 ELSE X=1004:RETURN

 1910 X=963:RETURN

 1920 N=INT(RND*3)+1:ON N GOTO 1930,1940,1970

 1930 RETURN

 1940 E=X-64:GOSUB 550:PRINT HP$:E=X-127:GOSUB 550:PRINT HO$

 1950 E=162 :GOSUB 550:PRINT HJ$:E=99   :GOSUB 550:PRINT HI$        ' 14

 1960 E=169 :GOSUB 550:PRINT HD$:E=106  :GOSUB 550:PRINT HC$:RETURN ' 15

 1970 E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$

 1980 E=176 :GOSUB 550:PRINT HB$:E=113  :GOSUB 550:PRINT HA$:RETURN ' 16

 1990 N=INT(RND*3)+1:ON N GOTO 2000,2010,2030

 2000 RETURN

 2010 E=X-64:GOSUB 550:PRINT HP$:E=X-127:GOSUB 550:PRINT HO$

 2020 E=176 :GOSUB 550:PRINT HJ$:E=113  :GOSUB 550:PRINT HI$:RETURN ' 16 

 2030 E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$

 2040 E=183 :GOSUB 550:PRINT HJ$:E=120  :GOSUB 550:PRINT HI$:RETURN ' 17

 2050 N=INT(RND*4)+1:ON N GOTO 2060,2070,2100,2120

 2060 RETURN

 2070 E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$        'R

 2080 E=546 :GOSUB 550:PRINT HJ$:E=483  :GOSUB 550:PRINT HI$        'F 23

 2090 E=562 :GOSUB 550:PRINT HB$:E=499  :GOSUB 550:PRINT HA$:RETURN 'L 25

 2100 E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$        'L

 2110 E=546 :GOSUB 550:PRINT HP$:E=483  :GOSUB 550:PRINT HO$:RETURN 'U 23

 2120 E=X-64:GOSUB 550:PRINT HR$:E=X-127:GOSUB 550:PRINT HQ$        'D

 2130 E=554 :GOSUB 550:PRINT HD$:E=491  :GOSUB 550:PRINT HC$:RETURN 'R 24

 2140 N=INT(RND*3)+1:ON N GOTO 2150,2160,2180

 2150 RETURN

 2160 E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$        'R

 2170 E=554 :GOSUB 550:PRINT HJ$:E=491  :GOSUB 550:PRINT HI$:RETURN 'F 24 

 2180 E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$        'L

 2190 E=562 :GOSUB 550:PRINT HP$:E=499  :GOSUB 550:PRINT HO$:RETURN 'U 25

 2200 N=INT(RND*6)+1:ON N GOTO 2210,2220,2240,2270,2300,2210

 2210 RETURN

 2220 E=X-64:GOSUB 550:PRINT HP$:E=X-127:GOSUB 550:PRINT HO$        'U

 2230 E=940 :GOSUB 550:PRINT HR$:E=877  :GOSUB 550:PRINT HQ$:RETURN 'D 33

 2240 IF X3<>0 THEN RETURN

 2250 E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$        'R

 2260 E=130 :GOSUB 550:PRINT HJ$:E=67   :GOSUB 550:PRINT HI$:RETURN 'F 10

 2270 IF X3<>0 THEN RETURN

 2280 E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$        'L

 2290 E=517 :GOSUB 550:PRINT HP$:E=454  :GOSUB 550:PRINT HO$:RETURN 'U 20

 2300 IF X3<>0 THEN RETURN

 2310 E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$        'R

 2320 E=899 :GOSUB 550:PRINT HJ$:E=836  :GOSUB 550:PRINT HI$:RETURN 'F 30

 2330 ' Print Executioners

 2340 E=X    :GOSUB 550:PRINT BA$:E=X+64 :GOSUB 550:PRINT BB$

 2350 E=X-64 :GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$

 2360 E=X+128:GOSUB 550:PRINT FC$:RETURN

 2370 GOSUB 450:E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$ 'R moving

 2380 GOSUB 450:E=X-64:GOSUB 550:PRINT HF$:E=X-127:GOSUB 550:PRINT HE$

 2390 GOSUB 450:E=X-64:GOSUB 550:PRINT HH$:E=X-127:GOSUB 550:PRINT HG$:RETURN

 2400 GOSUB 450:E=X-64:GOSUB 550:PRINT HF$:E=X-127:GOSUB 550:PRINT HE$ 'L moving

 2410 GOSUB 450:E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$

 2420 GOSUB 450:E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$:RETURN

 2430 GOSUB 460:E=X-64:GOSUB 550:PRINT HJ$:E=X-127:GOSUB 550:PRINT HI$ ' Left turn

 2440 GOSUB 460:E=X-64:GOSUB 550:PRINT HL$:E=X-127:GOSUB 550:PRINT HK$

 2450 GOSUB 450:E=X-64:GOSUB 550:PRINT HN$:E=X-127:GOSUB 550:PRINT HM$:RETURN

 2460 GOSUB 450:E=X-64:GOSUB 550:PRINT HL$:E=X-127:GOSUB 550:PRINT HK$ ' NO

 2470 GOSUB 450:E=X-64:GOSUB 550:PRINT HJ$:E=X-127:GOSUB 550:PRINT HI$

 2480 GOSUB 450:E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$:RETURN

 2490 ' Executioner No

 2500 FOR L=1 TO 3

 2510  GOSUB 450:E=X-64:GOSUB 550:PRINT HJ$:E=X-127:GOSUB 550:PRINT HI$

 2520  GOSUB 450:E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$

 2530  GOSUB 450:E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$

 2540  GOSUB 450:E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$

 2550 NEXT:RETURN

 2560 '

 2570 GOSUB 450:E=X-64:GOSUB 550:PRINT HP$:E=X-127:GOSUB 550:PRINT HO$:RETURN 'U

 2580 GOSUB 450:E=X-64:GOSUB 550:PRINT HR$:E=X-127:GOSUB 550:PRINT HQ$:RETURN 'D

 2590 ' Executioner Yes

 2600 FOR L=1 TO 3

 2610  GOSUB 370:GOSUB 450:E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$

 2620  GOSUB 380:GOSUB 450:E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$

 2630 NEXT:RETURN

 2640 GOSUB 450:E=X+4:GOSUB 550:PRINT BE$:E=X+68:GOSUB 550:PRINT BF$   ' Shooting hand

 2650 GOSUB 450:E=X+4:GOSUB 550:PRINT BG$:E=X+68:GOSUB 550:PRINT BH$

 2660 GOSUB 450:E=X+4:GOSUB 550:PRINT BI$:E=X+68:GOSUB 550:PRINT BJ$

 2670 GOSUB 450:E=X+4:GOSUB 550:PRINT BK$

 2680 GOSUB 450:E=X+4:GOSUB 550:PRINT BL$:RETURN                       ' Pistol

 2690 GOSUB 450:E=X+4:GOSUB 550:PRINT BK$

 2700 X3=X:GOSUB 430:GOSUB 430:X=X3

 2710 GOSUB 460:E=X+4 :GOSUB 550:PRINT BI$:E=X+68 :GOSUB 550:PRINT BJ$ ' Put RH down

 2720 GOSUB 450:E=X+4 :GOSUB 550:PRINT BG$:E=X+68 :GOSUB 550:PRINT BH$

 2730 GOSUB 450:E=X+4 :GOSUB 550:PRINT BE$:E=X+68 :GOSUB 550:PRINT BF$

 2740 GOSUB 460:E=X+4 :GOSUB 550:PRINT BC$:E=X+68 :GOSUB 550:PRINT BD$:RETURN

 2750 GOSUB 450:E=X-64:GOSUB 550:PRINT HD$:E=X-127:GOSUB 550:PRINT HC$:RETURN

 2760 GOSUB 450:E=X-64:GOSUB 550:PRINT HJ$:E=X-127:GOSUB 550:PRINT HI$:RETURN

 2770 GOSUB 450:E=X-64:GOSUB 550:PRINT HB$:E=X-127:GOSUB 550:PRINT HA$:RETURN

 2780 ' Animation of figures

 2790 M=INT(RND*16)+1:Y=INT(RND*18)+1:GOSUB 440

 2800 IF X3=X THEN 2790

 2810 IF X3=0 THEN 2830

 2820 IF (M<8) OR (M>12) THEN 2830 ELSE 2790

 2830 IF M>7 THEN 2990 ELSE ON M GOSUB 340,2430,2570,2580,2750,2760,2770

 2840 GOTO 2890

 2850 ON Y GOTO 

        2860,2870,2880,1740,1750,1760,1770,1780,1790,1800,1820,1830,1840,1850,1860,1880,1890,

        1900

 2860 X=194:RETURN

 2870 X=581:RETURN

 2880 X=963:RETURN

 2890 X1=INT(RND*5)+1

 2900 ON X1 GOTO 2910,2790,2910,2790,2910

 2910 ON M  GOTO 2970,2980,2920,2930,2940,2950,2960

 2920 E=X-127:GOSUB 550:PRINT H4$:GOSUB 450:E=X-127:GOSUB 550:PRINT HO$:RETURN

 2930 E=X-64 :GOSUB 550:PRINT H5$:GOSUB 450:E=X-64 :GOSUB 550:PRINT HR$:RETURN

 2940 E=X-64 :GOSUB 550:PRINT H2$:GOSUB 450:E=X-64 :GOSUB 550:PRINT HD$:RETURN

 2950 E=X-64 :GOSUB 550:PRINT H3$:GOSUB 450:E=X-64 :GOSUB 550:PRINT HJ$:RETURN

 2960 E=X-64 :GOSUB 550:PRINT H1$:GOSUB 450:E=X-64 :GOSUB 550:PRINT HB$:RETURN

 2970 E=X-64 :GOSUB 550:PRINT H6$:GOSUB 450:E=X-64 :GOSUB 550:PRINT HH$:RETURN

 2980 E=X-64 :GOSUB 550:PRINT H7$:GOSUB 450:E=X-64 :GOSUB 550:PRINT HN$:RETURN

 2990 ON M-7 GOTO 3000,3020,3040,3060,3090,3110,3130,3150,3160

 3000 E=67 :GOSUB 550:PRINT HO$:E=130:GOSUB 550:PRINT HP$                  ' 10

 3010 E=454:GOSUB 550:PRINT HC$:E=517:GOSUB 550:PRINT HD$:GOSUB 450:RETURN ' 20

 3020 E=67 :GOSUB 550:PRINT HQ$:E=130:GOSUB 550:PRINT HR$                  ' 10

 3030 E=836:GOSUB 550:PRINT HI$:E=899:GOSUB 550:PRINT HJ$:GOSUB 450:RETURN ' 30

 3040 E=454:GOSUB 550:PRINT HA$:E=517:GOSUB 550:PRINT HB$                  ' 20

 3050 E=836:GOSUB 550:PRINT HO$:E=899:GOSUB 550:PRINT HP$:GOSUB 450:RETURN ' 30

 3060 E=67 :GOSUB 550:PRINT HQ$:E=130:GOSUB 550:PRINT HR$                  ' 10

 3070 E=454:GOSUB 550:PRINT HI$:E=517:GOSUB 550:PRINT HJ$                  ' 20

 3080 E=836:GOSUB 550:PRINT HO$:E=899:GOSUB 550:PRINT HP$:GOSUB 450:RETURN ' 30

 3090 E=454:GOSUB 550:PRINT HO$:E=517:GOSUB 550:PRINT HP$                  ' 20

 3100 E=67 :GOSUB 550:PRINT HQ$:E=130:GOSUB 550:PRINT HR$:GOSUB 450:RETURN ' 10

 3110 IF (X=194) OR (X=581) OR (X=963) THEN 3240

 3120 E=X:GOSUB 550:PRINT BM$:E=X+64:GOSUB 550:PRINT BN$:GOSUB 450:RETURN

 3130 IF (X=194) OR (X=581) OR (X=963) THEN 3260

 3140 E=X  :GOSUB 550:PRINT BA$:E=X+64:GOSUB 550:PRINT BB$:GOSUB 450:RETURN

 3150 E=X+4:GOSUB 550:PRINT BO$:E=X+68:GOSUB 550:PRINT BP$:GOSUB 450:RETURN

 3160 E=X+4:GOSUB 550:PRINT BG$:E=X+68:GOSUB 550:PRINT BH$:GOSUB 450:RETURN

 3170 ' Slow down                                                      PowerBasic

 3180 DELAY=550/100000!    :START=

        TIMER:WHILE TIMER<START+DELAY:WEND:RETURN ' DELAY 1/13:RETURN

 3190 DELAY=430/100000!:START=TIMER:WHILE TIMER<START+DELAY:WEND:RETURN ' DELAY 1/15:RETURN

 3200 DELAY=1          :START=TIMER:WHILE TIMER<START+DELAY:WEND:RETURN ' DELAY 1/17:RETURN

 3210 DELAY=230/10000:START=TIMER:WHILE TIMER<START+DELAY:WEND:RETURN ' DELAY 1/19:RETURN

 3220 DELAY=130/10000:START=TIMER:WHILE TIMER<START+DELAY:WEND:RETURN ' DELAY 1/30:RETURN

 3230 FOR O=1 TO 3:GOSUB 430:NEXT:RETURN

 3240 E=X   :GOSUB 550:PRINT BR$

 3250 E=X+64:GOSUB 550:PRINT BN$:GOSUB 450:RETURN

 3260 E=X   :GOSUB 550:PRINT BQ$

 3270 E=X+64:GOSUB 550:PRINT BB$:GOSUB 450:RETURN

 3280 ' Your turn, X9=1    My turn, X9=2

 3290 LSET Q=MKI$(0)

 3300 WHILE CVI(Q)=0

 3310 X8=0:IF X9=2 THEN 850 ELSE X3=0:E=22:GOSUB 550:PRINT "*":GOSUB 430

 3320 MID$(Q,1)=INKEY$:IF CVI(Q) THEN 3330

 3330 WEND                                           ' check which row

 3340 IF ASC(Q)=27 THEN 4910                         ' exits game

 3350 IF (ASC(Q)>48) AND (ASC(Q)<52) THEN 3370

 3360 IF ASC(Q)=114 OR ASC(Q)=82 THEN 1380 ELSE 3290 ' r R

 3370 RW=VAL(Q):ON RW GOSUB 500,3640,3650            ' print 1-3

 3380 GOSUB 570                                      ' check 1-7 remove

 3390 IF (ASC(Q)>48) AND (ASC(Q)<56) THEN 3420

 3400 IF ASC(Q)=32 THEN GOSUB 490 ELSE 3380          ' space restarts

 3410 GOTO 3290

 3420 X9=2                                           ' computer's turn

 3430 RN$=Q:RN=VAL(RN$)

 3440 IF (RN>VAL(RW$(RW))) AND (X9=1) THEN X8=1

 3450 IF (RN>VAL(RW$(RW))) AND (X9=2) THEN X9=1

 3460 ON RW GOSUB 3490,3500,3510                     ' print amount to remove

 3470 IF(VAL(RW$(1))=0) AND (VAL(RW$(2))=0) AND (VAL(RW$(3))=0) THEN 1240 ' end of game

 3480 IF X8=0 THEN 3290 ELSE 1120

 3490 E=128:GOSUB 550:PRINT RN$:X=194:GOTO 3530

 3500 E=512:GOSUB 550:PRINT RN$:X=581:GOTO 3530

 3510 E=896:GOSUB 550:PRINT RN$:X=963

 3520 ' Check if amount is OK

 3530 GOSUB 360:X3=X:GOSUB 430:X=X3:GOSUB 420:GOSUB 340:X3=X:GOSUB 430:X=X3:GOSUB 350

 3540 IF RN>VAL(RW$(RW)) THEN 3550 ELSE 3610

 3550 GOSUB 2490                                     ' Executioner NO

 3560 ON RW GOTO 3570,3580,3590                      ' Clear score

 3570 E=128:GOSUB 550:PRINT " ":E=192:GOSUB 550:PRINT " ":RETURN

 3580 E=512:GOSUB 550:PRINT " ":E=576:GOSUB 550:PRINT " ":RETURN

 3590 E=896:GOSUB 550:PRINT " ":E=960:GOSUB 550:PRINT " ":RETURN

 3600 ' Executioner OK and Shoots

 3610 GOSUB 390:GOSUB 340:X3=X:GOSUB 430:X=X3:GOSUB 400

 3620 GOSUB 530:GOSUB 540:GOSUB 410:GOSUB 350:GOTO 3560

 3630 E=192:GOSUB 550:PRINT "1":RETURN

 3640 E=576:GOSUB 550:PRINT "2":RETURN

 3650 E=960:GOSUB 550:PRINT "3":RETURN

 3660 ' Faces turn to the left and look up or down

 3670 ON RW GOTO 3680,3720,3760

 3680 X=254:IF VAL(RW$(1))=0 THEN 3690 ELSE FOR O=1 TO VAL(RW$(1)):X=X-7:GOSUB 360:NEXT

 3690 X=581:GOSUB 370:X=634 :IF VAL(RW$(2))=0 THEN 3700 

        ELSE FOR O=1 TO VAL(RW$(2)):X=X-8:GOSUB 370:NEXT

 3700 X=963:GOSUB 370:X=1013:IF VAL(RW$(3))=0 THEN 3710 

        ELSE FOR O=1 TO VAL(RW$(3)):X=X-9:GOSUB 370:NEXT

 3710 X=194:RETURN

 3720 X=634:IF VAL(RW$(2))=0 THEN 3730 ELSE FOR O=1 TO VAL(RW$(2)):X=X-8:GOSUB 360:NEXT

 3730 X=194:GOSUB 380:X=254 :IF VAL(RW$(1))=0 THEN 3740 

        ELSE FOR O=1 TO VAL(RW$(1)):X=X-7:GOSUB 380:NEXT

 3740 X=963:GOSUB 370:X=1013:IF VAL(RW$(3))=0 THEN 3750 

        ELSE FOR O=1 TO VAL(RW$(3)):X=X-9:GOSUB 370:NEXT

 3750 X=581:RETURN

 3760 X=1013:IF VAL(RW$(3))=0 THEN 3770 ELSE FOR O=1 TO VAL(RW$(3)):X=X-9:GOSUB 360:NEXT

 3770 X=581 :GOSUB 380:X=634:IF VAL(RW$(2))=0 THEN 3780 

        ELSE FOR O=1 TO VAL(RW$(2)):X=X-8:GOSUB 380:NEXT

 3780 X=194 :GOSUB 380:X=254:IF VAL(RW$(1))=0 THEN 3790 

        ELSE FOR O=1 TO VAL(RW$(1)):X=X-7:GOSUB 380:NEXT

 3790 X=963:RETURN

 3800 ' Shoot Androids

 3810 Z1=VAL(RW$(RW)):Z2=Z1-RN:RW$(RW)=STR$(Z1-RN)

 3820 ON RW GOTO 3830,3840,3850

 3830 Z$="===-":Z3=205:Z4=7:Z5=254 -7*Z1:GOTO 3860

 3840 Z$="===-":Z3=592:Z4=8:Z5=634 -8*Z1:GOTO 3860

 3850 Z$="===-":Z3=974:Z4=9:Z5=1013-9*Z1

 3860 FOR M=1 TO RN

 3870  E=Z3-1:GOSUB 550:PRINT "="

 3880   FOR N=Z3 TO Z5 STEP 3

 3890    E=Z3-1:GOSUB 550:PRINT "-"

 3900    E=N:GOSUB 550:PRINT Z$

 3910    E=Z3-1:GOSUB 550:PRINT "="

 3920   NEXT

 3930  E=Z5    :GOSUB 550:PRINT FA$   ' white blast

 3940  E=Z5+64 :GOSUB 550:PRINT FA$

 3950  E=Z5-64 :GOSUB 550:PRINT FA$

 3960  E=Z5+128:GOSUB 550:PRINT FA$

 3970  E=Z5-128:GOSUB 550:PRINT FA$

 3980  E=Z5    :GOSUB 550:PRINT FB$   ' blank androids

 3990  E=Z5+64 :GOSUB 550:PRINT FB$

 4000  E=Z5-64 :GOSUB 550:PRINT FB$

 4010  E=Z5+128:GOSUB 550:PRINT FB$

 4020  E=Z5-128:GOSUB 550:PRINT FB$

 4030   FOR N=Z3 TO Z5 STEP 3

 4040    E=N:GOSUB 550:PRINT "    "

 4050   NEXT

 4060  Z5=Z5+Z4

 4070 NEXT

 4080 Y9=Y9+1:X3=X:GOSUB 430:X=X3:RETURN

 4090 ' Print @ to Locate converter and Color figures

 4100 V=INT(E/64)+1:F=(V-1)*64:F=(E-F)+1:LOCATE V,F

 4110 IF E<26  THEN COLOR 13,0:GOTO 4150 ' heading light magenta

 4120 IF E<400 THEN COLOR 14,0:GOTO 4150 ' row 1   yellow

 4130 IF E>800 THEN COLOR 12,0:GOTO 4150 ' row 3   light red

 4140 COLOR 10,0                         ' row 2   light green

 4150 RETURN

 4160 ' Instructions

 4170 COLOR 10,0

 4180 PRINT "ANDROID NIM: TRS-80 BASIC BY LEO CHRISTOPHERSON"

 4190 PRINT "GwBasic by Eric F. Tchong - Aruba - August 1996":COLOR 11,0:PRINT

 4200 PRINT "When you see an (*) to the left of 'ANDROID NIM'"

 4210 PRINT "its your turn.":PRINT

 4220 PRINT "You need not push enter, just the row number, 1"

 4230 PRINT "2 or 3,  and the number of androids you wish to"

 4240 PRINT "have removed. To give up, push the  <R/r>  key."

 4250 PRINT:COLOR 12,0

 4260 PRINT "RULES: You may remove as many androids from any"

 4270 PRINT "row as you wish when its your turn.  To win you"

 4280 PRINT "must remove the last android.":PRINT:COLOR 13,0

 4290 PRINT "When its your turn press <Esc> to exit program.":COLOR 14,0

 4300 PRINT:INPUT "Push <Enter> when ready ";X

 4310 ' Build strings

 4320 HA$=CHR$(220)+CHR$(221)+CHR$(222)+CHR$(220)

 4330 HB$=CHR$(255)+CHR$(220)+CHR$(223)+CHR$(220)+CHR$(219)+CHR$(255)

 4340 H1$=CHR$(255)+CHR$(219)+CHR$(223)+CHR$(219)+CHR$(254)

 4350 HC$=CHR$(220)+CHR$(221)+CHR$(220)+CHR$(221)

 4360 HD$=CHR$(255)+CHR$(219)+CHR$(220)+CHR$(223)+CHR$(220)+CHR$(255)

 4370 H2$=CHR$(255)+CHR$(219)+CHR$(219)+CHR$(223)+CHR$(219)

 4380 HE$=CHR$(220)+CHR$(220)+CHR$(221)+CHR$(221)

 4390 HF$=CHR$(255)+CHR$(219)+CHR$(219)+CHR$(220)+CHR$(223)+CHR$(220)+CHR$(255)

 4400 HG$=CHR$(220)+CHR$(220)+CHR$(221)+CHR$(220)

 4410 HH$=CHR$(255)+CHR$(219)+CHR$(219)+CHR$(220)+CHR$(219)+CHR$(220)+CHR$(255)

 4420 H6$=CHR$(255)+CHR$(219)+CHR$(219)+CHR$(219)+CHR$(219)+CHR$(220)

 4430 HI$=CHR$(222)+CHR$(220)+CHR$(222)+CHR$(220)

 4440 HJ$=CHR$(222)+CHR$(220)+CHR$(223)+CHR$(220)+CHR$(219)+CHR$(255)

 4450 H3$=CHR$(255)+CHR$(219)+CHR$(223)+CHR$(219)+CHR$(255)

 4460 HK$=CHR$(222)+CHR$(220)+CHR$(221)+CHR$(220)

 4470 HL$=CHR$(220)+CHR$(223)+CHR$(220)+CHR$(219)+CHR$(219)+CHR$(255)

 4480 HM$=CHR$(220)+CHR$(221)+CHR$(220)+CHR$(220)

 4490 HN$=CHR$(220)+CHR$(221)+CHR$(220)+CHR$(219)+CHR$(219)+CHR$(255)

 4500 H7$=CHR$(220)+CHR$(219)+CHR$(219)+CHR$(219)+CHR$(219)

 4510 BA$=CHR$(220)+CHR$(223)+CHR$(219)+CHR$(219)+CHR$(223)+CHR$(220)

 4520 BB$=CHR$(223)+CHR$(220)+CHR$(223)+CHR$(223)+CHR$(220)+CHR$(223)

 4530 FC$=CHR$(220)+CHR$(220)+CHR$(221)+CHR$(222)+CHR$(220)+CHR$(220)

 4540 HO$=CHR$(223)+CHR$(220)+CHR$(220)+CHR$(223)

 4550 HP$=CHR$(255)+CHR$(219)+CHR$(220)+CHR$(220)+CHR$(219)+CHR$(255)

 4560 H4$=CHR$(220)+CHR$(221)+CHR$(222)+CHR$(220)

 4570 HQ$=CHR$(220)+CHR$(221)+CHR$(222)+CHR$(220)

 4580 HR$=CHR$(255)+CHR$(223)+CHR$(219)+CHR$(223)+CHR$(219)+CHR$(255)

 4590 H5$=CHR$(255)+CHR$(219)+CHR$(223)+CHR$(219)+CHR$(223)

 4600 BC$=CHR$(223)+CHR$(220)

 4610 BD$=CHR$(220)+CHR$(223)

 4620 BE$=CHR$(219)+CHR$(255)

 4630 BF$=CHR$(223)+CHR$(255)

 4640 BG$=CHR$(219)+CHR$(255)+CHR$(255)

 4650 BH$=CHR$(223)+CHR$(220)

 4660 BI$=CHR$(219)+CHR$(220)+CHR$(220)+CHR$(255)+"  "

 4670 BJ$=CHR$(255)+CHR$(255)

 4680 BK$=CHR$(223)+CHR$(220)+CHR$(220)+",-= "

 4690 BL$=CHR$(220)+STRING$(3,CHR$(254))+"'-="

 4700 BM$=CHR$(255)+CHR$(219)+CHR$(219)+CHR$(223)+CHR$(219)+CHR$(255)

 4710 BN$=CHR$(220)+CHR$(223)+CHR$(221)+CHR$(222)+CHR$(220)+CHR$(223)

 4720 BO$=CHR$(223)+CHR$(220):BP$=CHR$(220)+CHR$(223)

 4730 FA$=STRING$(6,CHR$(219)):FB$=STRING$(6,CHR$(255))

 4740 BR$=CHR$(255)+CHR$(219)+CHR$(219)+CHR$(219)+CHR$(223)+CHR$(220)

 4750 BQ$=CHR$(220)+CHR$(223)+CHR$(219)+CHR$(219)+CHR$(223)+CHR$(220)

 4760 MA$="      █  █     █  █  ███  █"

 4770 MB$="      █  █  ▄  █  █  █ ██ █"

 4780 MC$="      █   █▄█▄█   █  █  ███"

 4790 GOTO 590

 4800 ' Wait for key

 4810 LSET Q=MKI$(0)

 4820 WHILE CVI(Q)=0:MID$(Q,1)=INKEY$:WEND:RETURN

 4830 ' Wait for key and animate

 4840 LSET Q=MKI$(0)

 4850 WHILE CVI(Q)=0

 4860  GOSUB 430

 4870  MID$(Q,1)=INKEY$:IF CVI(Q) THEN 4880

 4880 WEND

 4890 RETURN

 4900 ' Exit

 4910 KEY 5,"droidnim.bas":KEY 6,CHR$(34)+",a"

 4920 COLOR 7,0:CLS:KEY LIST

